# Identifiability of an SIR model

## Model Equations

(moved the derivatives to the right hand side, so each equation = 0)

```
Seqn[t_] = μ NN - β S[t] II[t] - μ S[t] - D[S[t], t]
Ieqn[t_] = β S[t] II[t] - (μ + γ) II[t] - D[II[t], t]
```

$$NN\,\mu - \mu\,S[t] - \beta\,II[t]\,S[t] - S'[t]$$

$$-(\gamma + \mu)\,II[t] + \beta\,II[t]\,S[t] - II'[t]$$

## Substitute I = y/k into the I equation

```
yeqn[t_] = Ieqn[t] /. {II[t] → y[t] / k, II'[t] → y'[t] / k}
```

$$-\frac{(\gamma+\mu)\,y[t]}{k} + \frac{\beta\,S[t]\,y[t]}{k} - \frac{y'[t]}{k}$$

## Solve for S

```
Solve[yeqn[t] == 0, S[t]]
```

$$\left\{\left\{S[t] \to \frac{\gamma\,y[t] + \mu\,y[t] + y'[t]}{\beta\,y[t]}\right\}\right\}$$

```
D[Solve[yeqn[t] == 0, S[t]], t]
```

$$\left\{\left\{S'[t] \to -\frac{y'[t]\,(\gamma\,y[t] + \mu\,y[t] + y'[t])}{\beta\,y[t]^2} + \frac{\gamma\,y'[t] + \mu\,y'[t] + y''[t]}{\beta\,y[t]}\right\}\right\}$$

## Plug in to the S equation and clear denominators to get our input - output equation

```
AlmostIPOP[t_] = Seqn[t] /. Solve[yeqn[t] == 0, S[t]][[1]] /.
   D[Solve[yeqn[t] == 0, S[t]], t][[1]] /. {II[t] → y[t] / k}
```

$$NN\,\mu - \frac{\gamma\,y[t] + \mu\,y[t] + y'[t]}{k} - \frac{\mu\,(\gamma\,y[t] + \mu\,y[t] + y'[t])}{\beta\,y[t]} + \frac{y'[t]\,(\gamma\,y[t] + \mu\,y[t] + y'[t])}{\beta\,y[t]^2} - \frac{\gamma\,y'[t] + \mu\,y'[t] + y''[t]}{\beta\,y[t]}$$

```
IPOP[t_] = Denominator[Together[AlmostIPOP[t]]] Together[AlmostIPOP[t]]
```

$k\,NN\,\beta\,\mu\,y[t]^2 - k\,\gamma\,\mu\,y[t]^2 - k\,\mu^2\,y[t]^2 - \beta\,\gamma\,y[t]^3 - \beta\,\mu\,y[t]^3 - k\,\mu\,y[t]\,y'[t] - \beta\,y[t]^2\,y'[t] + k\,y'[t]^2 - k\,y[t]\,y''[t]$

# Next we collect common terms and make the equation monic. The monomial terms of the input-output equation are:

```
MonomialList[IPOP[t] / k, {y[t], y'[t], y''[t]}]
```

$\left\{\left(-\frac{\beta\,\gamma}{k} - \frac{\beta\,\mu}{k}\right) y[t]^3,\ -\frac{\beta\,y[t]^2\,y'[t]}{k},\ \left(NN\,\beta\,\mu - \gamma\,\mu - \mu^2\right) y[t]^2,\ -\mu\,y[t]\,y'[t],\ -y[t]\,y''[t],\ y'[t]^2\right\}$

Here you can see the coefficents of the terms and how they lead us to having $\mu$, $\gamma$, $\beta$/k, and $\beta$ N as the identifiable combinations.